

# Summing It up with Riemann, Definite Integrals, and the Fundamental Theorem of Calculus

*Note: You may notice differences between this Maple worksheet and the equivalent Mathematica notebook. These differences were introduced to preserve the content of these modules and were necessary because of major functional differences between Maple and Mathematica.*

## Introduction

**OBJECTIVE:** Visualize the relationship among Riemann sums, the definite integral, and the Fundamental Theorem of the integral calculus. Use each of the concepts to compute or approximate the signed area under the graph of a function representing various applications.

In practical applications, the signed area between the graph of a function and the  $x$  or  $t$ -axis can represent a wide variety of different things. For example, if a function gives the rate of water flowing into a storage tank as a function of time, the total signed area between the graph of the flow-rate function and the  $t$ -axis represents the total amount of water that accumulates or drains from the tank during the time interval. To see that this is the case, you should recall something that you have probably known for a long time, that is, change in a quantity is equal to its rate of change times time. Some familiar examples of this are "distance is equal to speed times time" and "amount is equal to rate times time." If water flows into a tank at a constant rate of 30 liters per second for 100 seconds, then it is easy to determine that the total water added to the tank during this time is  $30 \times 100 = 3000$  liters. These simple rules are based, of course, on the assumption that the rate of change is constant during the time interval. Now that we are more sophisticated, we know that rates of change do not have to be constant and can, in fact, vary continuously over a time interval. How do we calculate the total change over a time interval when the rate varies from one instant to the next? Calculus provides the answer, and we begin by looking at a technique for estimating change from rates of change using a summation process called Riemann sums.

## Technology Guidelines

**NOTE:** If you have just finished a worksheet, **restart** *Maple* before executing a new worksheet.

**TO OPEN SECTIONS,**

Click on the **PLUS** sign at the left hand side of the screen *or* select **Expand All Sections** from the **View** drop down menu.

**TO STOP AN EXECUTION**

Click on **STOP** button from the toolbar.

**ORDER OF EXECUTION**

Execute commands in the order given. Do not skip any *Maple* Input lines within a given worksheet

Alternatively, you can execute the entire worksheet by selecting the **Execute Worksheet** command from the **Edit** drop down menu.

## SAVING WORKSHEETS.

You can save anytime to any directory you choose, and it is wise to save often.  
**EXPERIENCING MAJOR PROBLEMS**  
 Save if appropriate, and then shut down *Maple* and start it up again.

## Part I: An Example

The idea of Riemann sums builds upon what you already know, that is change is equal to rate of change times time. Suppose that  $f(t)$  represents the rate of change of some quantity  $Q$ , that is,

$f(t) = \frac{dQ}{dt}$ . The quantity  $Q$  might represent the amount of water in a reservoir, the distance


traveled by a moving object, or the amount of money in an investment portfolio. In these types of practical applications, it is often times easier to consider the forces that influence change and thereby determine  $f(t)$  than it is to determine  $Q$  directly. But if you know  $f(t)$ , there is a way to estimate the change in  $Q$  over an interval of time, even though the rate of change varies from one instant to the next. Here it is.

Let's consider a specific example. Suppose that  $f(t) = \frac{1}{4} - \cos\left(\frac{\pi t}{12}\right)$  for  $0 \leq t \leq 15$ , and that

we wish to calculate  $\Delta Q$  over this time interval. First, let's graph  $f(t)$ .

- > **restart;**  
**with(student):**
- > **f := t -> 1/4 - cos(Pi\*t/12);**  
**plot( f(t), t=0..15, labels=["t", "y = f(t)"], thickness = 2, color=green );**

$$f := t \rightarrow \frac{1}{4} - \cos\left(\frac{1}{12}\pi t\right)$$



We know from the definition of differentials that over a short interval of time  $\Delta Q$  is

approximately equal to  $dQ = \frac{dQ}{dt} dt = f(t) dt$ . This suggests a method of attack. If we partition

the 15-hour period into very short time intervals, called subintervals, and then approximate the change in  $Q$  over each of these subintervals and add them up, we should get an approximation of the total  $\Delta Q$  as  $t$  goes from 0 to 15. We let  $dt$  be the length or duration of each of the

subintervals and evaluate  $f(t)$  at some instant in each subinterval. Where should we pick  $t$  in each subinterval? If the subintervals are short enough, it shouldn't make too much difference. (You might guess, however, that it would probably be better to pick  $t$  somewhere near the middle of each subinterval since that would most likely give a value of  $f(t)$  that is near its average value over each subinterval.)

We can say all of this mathematically as follows. First we divide the total 15-hour period into  $n$  subintervals each of length  $dt$  and number each subinterval from 1 to  $n$ . Then,

$$\Delta Q = \sum_{i=1}^n \Delta Q_i \quad \text{which is approximately equal to} \quad \sum_{i=1}^n dQ_i = \sum_{i=1}^n f(c_i) h, \quad \text{where } h = dt \text{ and}$$

$c_i$  is any value of  $t$  taken from the  $i^{\text{th}}$  subinterval. The sum that we form in this way to

approximate  $\Delta Q$  is called a Riemann sum.

There is a geometric view that can be attached to the Riemann sum,  $\sum_{i=1}^n f(c_i) h$ . The product,


$f(c_i) h$ , can be thought of as the area of a rectangle that is  $h$  units wide and  $f(c_i)$  units high. In each subinterval, we can draw this rectangle that is  $h$  units wide and  $f(c_i)$  units high. The base of each rectangle is on the  $x$  or  $t$  axis, and the opposite side is on the graph of  $f(t)$ . The Riemann sum is the sum of the areas of these rectangles. If we admit that  $f(c_i)$  could be negative, then the area of a rectangle can be negative; hence we add signed areas in a Riemann sum, which itself can be positive, negative, or 0. The product can be negative when  $f(c_i)$  is positive and  $h$  is negative.

This would occur if we were to sum back through the time history of  $f(t)$ . It makes sense that if  $Q$  increases as time progresses, then it would decrease if were to go backwards through the same time interval, and vice versa.

To illustrate the geometry of Riemann sums, we use the *Maple* **leftbox()** and **rightbox()** functions from the **student** package. The arguments of **leftbox()** and **rightbox()** are the function **f(t)**, the independent variable **t**, and a plot range.

Let's use **leftbox()** on the function we have been considering and use 15 rectangles.

> **leftbox( f(t), t=0..15, 15);**



Note that the rectangles under the  $t$  axis have negative areas, whereas the ones above have positive areas. Try changing the number of rectangles in the **leftbox()** command to see that the total signed area of the rectangles gets closer to the signed area under the graph of  $f(t)$ . You can also

change the function and the bounds on the interval to explore other possibilities if you wish.

## You Try It: Part I

In addition to **leftbox()**, *Maple* has written **rightbox()** and **middlebox()** commands so that you can see the geometry of a Riemann sum when we pick points other than the left endpoint in each subinterval. The arguments are the same as for **leftbox()**. Use these commands to visualize the Riemann sums for the function considered in the preceding section and for some other functions that you choose. Here is an example. This may take a few minutes to execute.

First, try it on the function we have selected, then try it with one of your own by changing the appropriate entries in the commands that follow. Be careful to use correct notation when inserting a function.


```
> f := t -> -3*t^3;
  a := -2;
  b := 2;
  n := 20;
  leftbox( f(t), t=a..b, n, title="Left Reimann Sum" , labels=["t", "y"]);
  middlebox( f(t), t=a..b, n, title="Middle Reimann Sum", labels=["t", "y"] );
  rightbox( f(t), t=a..b, n, title="Right Reimann Sum", labels=["t", "y"]);
```



$$f := t \rightarrow -3t^3$$

$$a := -2$$

$$b := 2$$

$$n := 20$$





Geometrically, which looks as if it would give the best results? Try it with another function.

## Part II: Summing It Up with Riemann

We need to calculate the total signed area of the rectangles. First we calculate  $h$  and  $c_i$  for each of the subintervals, and then we add up the signed areas of the rectangles. We use the left endpoint in each interval for  $c_i$ .

```

> f := t -> 1/4 - cos(Pi*t/12):
a := 0:
b := 15:
n := 15:
area := leftsum( f(t), t=a..b, n ):
printf( `The total signed area of the rectangles is %a`, evalf( area ) );

```

The total signed area of the rectangles is 5.581951230


To see what happens as we increase the number of rectangles, we combine the **leftbox** ( ) command together with the commands in the preceding section and put them all in a single section, the one that follows. With this we can easily change **f(t)**, **n**, **a**, and/or **b**.

```

> f := t -> 1/4 - cos(Pi*t/12):
a := 0:
b := 15:
n := 30:
area := leftsum( f(t), t=a..b, n ):
printf( `The total signed area of the rectangles is %a`, evalf( area ) );
leftbox( f(t), t=a..b, n );

```

The total signed area of the rectangles is 6.020314480




If we let the number of subintervals and corresponding rectangles approach infinity, we might expect the Riemann sum to approach the actual signed area under the graph of  $f(t)$ , provided  $f$  is continuous on  $[a, b]$ , and that this value will be exactly equal to  $\Delta Q$  over the interval.

## You Try It: Part II

Try out your own function by changing the definition of  $f$ . Be careful to use the correct notation. Replace any of the appropriate terms with functions and domains of your own choosing.

```
> f := t -> exp(-t^2/2)/sqrt(2*Pi):
a := -2:
b := 2:
n := 30:
area := leftsum( f(t), t=a..b, n ):
printf( `The total signed area of the rectangles is %a`, evalf( area ) );
leftbox( f(t), t=a..b, n , view=[-3..3,0..0.5],labels=["t","y"]);
```

The total signed area of the rectangles is .9541798846



## Part III: Take it To the Limit - The Definite Integral

To evaluate the limit of the Riemann sum, we form the sum without assigning a specific value to  $n$  and then take the limit as  $n$  goes to infinity. In the first cell that follows, we use *Maple* to find an expression for the Riemann sum in terms of  $n$ , and in the next cell we take the limit.

```
> unassign('n'):
f := t -> 1/4 - cos(Pi*t/12):
a := 0:
b := 15:
area := leftsum( f(t), t=0..15, n ):
area := simplify( value( area ), 'trig' );
```



$$area := -\frac{15}{4} \frac{-n \sin\left(\frac{5\pi}{4n}\right) - \sqrt{2} \cos\left(\frac{5\pi}{4n}\right) - \sqrt{2} + \sqrt{2} \sin\left(\frac{5\pi}{4n}\right) + 2 \sin\left(\frac{5\pi}{4n}\right)}{n \sin\left(\frac{5\pi}{4n}\right)}$$

Now take the limit.

> **limit( area, n = infinity );**

$$\frac{15\pi + 24\sqrt{2}}{4\pi}$$

This is the exact value for the signed area under the graph of  $f(t)$ , and hence it is also the exact value for  $\Delta Q$ . The limit of a Riemann sum is called a definite integral and is specified by the

following notation,  $\int_a^b f(t) dt = \lim_{n \rightarrow \infty} \sum_{i=1}^n f(c_i) h$  where  $h = \frac{b-a}{n}$ . We can use *Maple* to

evaluate definite integrals as in the following cell.

> **deltaQ := int( f(t), t=0..15 );**

$$deltaQ := \frac{3(5\pi + 8\sqrt{2})}{4\pi}$$

Comparing this result with the value of the Riemann sum calculated above, we see that they are the same.

It is important to remember that a definite integral is defined as the limit of a Riemann sum. We are sure that the definite integral of a function will exist (i.e., the limit of the Riemann sum will be a finite number), provided that  $f(t)$  is continuous on the closed interval over which we perform the summation.

## You Try It: Part III

### Leftbox Approximations

For each of the following functions, use the group of commands copied into the section below to

estimate the definite integral using a Riemann sum. First use the **leftbox( )** command to show the graph and some rectangles (use fewer than 50 rectangles for a clear image). Then use the commands in the next cell to see if you can make an estimate of the definite integral that is accurate to 3 digits by varying  $n$ . Make note of how many subintervals you need to use to do this. After you make an estimate, use the definite integral command to see *Maple's* decimal estimate of the exact value, and compare it with yours. Here are some functions to try. We show (c) as an example and find by trial and error that 5836 rectangles are needed to get three-digit accuracy.

a)  $f(t) = 3t^2 - 6$  for  $(-1 \leq t) \leq 3$

b)  $f(t) = 2 \sin(t)$  for  $(-\pi \leq t) \leq 2\pi$


c)  $f(t) = \frac{2}{t^2}$  for  $(.5 \leq t) \leq 3$

d)  $f(t) = \sqrt{t}$  for  $(1 \leq t) \leq 4$

e)  $f(t) = \frac{1}{t}$  for  $(1 \leq t) \leq 2$

You can change the appropriate terms below to explore the functions listed above.

```
> f := t -> 2/t^2:
a := 0.5:
b := 3:
n := 15:
leftbox( f(t), t=a..b, n , view=[-1..3,0..10]);
```



```
> n := 5836;
   area := leftsum( f(t), t=a..b, n );
   printf( `The total signed area of the rectangles is %a`, evalf( area ) );
```

$n := 5836$

$$area := 0.0004283755997 \left( \sum_{i=0}^{5835} \left( \frac{2}{(0.5 + 0.0004283755997 i)^2} \right) \right)$$

The total signed area of the rectangles is 3.334999726

```
> evalf(int(f(t), t=a..b));
```

3.333333333

### Middlebox Approximations

Write a series of commands to calculate the Riemann sum using the midpoint of each subinterval. Use your series of commands to repeat the problems in exercise 1. Compare the number of rectangles that are required to get three-digit accuracy using the left Riemann sum and the midpoint Riemann sum, and discuss what you find. Use the command **middlebox()** to visualize the midpoint rectangles if you wish.

## Part IV: Sometimes There is an Easier Way - The Fundamental Theorem of Calculus

Remember that our motivation for the area problem is to calculate  $\Delta Q$  over an interval of its domain  $(a \leq t) \leq b$ , given that we know its derivative function,  $\frac{dQ}{dt} = f(t)$ . Our approach has been:

1. to approximate  $\Delta Q$  by dividing the interval of the domain into a sequence of subintervals
2. to then approximate the  $\Delta Q_i$ 's over each of these with  $dQ_i = f(c_i) h$  (i.e., over a short interval, change is approximately equal to rate times time)
3. to finally adding them up in a Riemann sum and take the limit as the number of subintervals goes to infinity.

As the number of subintervals increases, the approximation improves, approaching the exact  $\Delta Q$ . That is, as  $n$  goes to infinity, the limit of the Riemann sum gives the exact value of  $\Delta Q$ , and this we call the definite integral of  $f(t)$  over the interval  $[a, b]$ .

You are probably wondering, why all the fuss? Wouldn't it be easier simply to find  $Q = F(t)$ , an antiderivative of  $f(t)$ , and calculate  $\Delta Q = Q[t=b] - Q[t=a] = F(b) - F(a)$ ? Yes, in fact we can do exactly that. This can make the whole business of evaluating a definite integral a lot easier, provided there is a simple formula for the antiderivative, which isn't always the case. Let's do it

for the integral  $\int_0^{15} f(t) dt$ , where  $f(t) = \frac{1}{4} - \cos\left(\frac{\pi}{12}t\right)$ .

> **f := t -> 1/4 - cos(Pi\*t/12);**

$$f := t \rightarrow \frac{1}{4} - \cos\left(\frac{1}{12}\pi t\right)$$

First, we find an antiderivative of  $f(t)$  and then calculate  $\Delta Q$ .

> **F:=unapply(int(f(t), t)+C,t);**

$$F := t \rightarrow \frac{1}{4}t - \frac{12 \sin\left(\frac{1}{12}\pi t\right)}{\pi} + C$$

> **deltaQ := (a,b)->F(b)-F(a);**

$$\text{delta}Q := (a, b) \rightarrow F(b) - F(a)$$

> **deltaQ(0, 15);**

$$\frac{15}{4} + \frac{6\sqrt{2}}{\pi}$$

This is the same value we obtained by taking the limit of a Riemann sum! This result is *fundamental*. It tells us that we can evaluate the definite integral of a function,  $f(t)$ , either by 1) taking the limit of a Riemann sum or 2) finding an antiderivative of  $f(t)$ , let's call it  $F(t)$ , and taking the difference of this function over the interval of summation. We can summarize this as

follows:  $\int_a^b f(t) dt = \lim_{n \rightarrow \infty} \sum_{i=1}^n f(c_i) h = F(b) - F(a)$  or, in terms of  $Q$  and  $\frac{dQ}{dt}$ ,

$$\int_a^b \frac{dQ}{dt} dt = Q(b) - Q(a) .$$

This all makes good sense if you think of it as a generalization of our

old friend, change is equal to rate times time. This result is so important that it is called the Fundamental Theorem of Calculus, Part 2. The theorem assures us that we can evaluate a definite integral in this simpler way whenever  $f(t)$  is continuous over the interval of summation and provided a formula exists for the antiderivative of  $f(t)$ .

The Fundamental Theorem, Part 2, helps us in two ways. First, as we have discovered, it tells us that instead of taking the limit of a Riemann sum, we can evaluate the definite integral of a function  $f(t)$  by taking the difference of values of an antiderivative  $F(t)$  over the interval of integration, provided there is a formula for  $F(t)$ . But what if, as is often the case, there is no algebraic formula for  $F(t)$ ? Well, now we can turn the problem around. Even when we don't have a formula for  $F(t)$ , we can evaluate a change in the value of  $F(t)$  over an interval  $[a, b]$  by taking the limit of a Riemann sum. Specifically,

$$F(b) - F(a) = \int_a^b f(t) dt = \lim_{h \rightarrow 0} \sum_{i=1}^n f(c_i) h .$$

This is the second way that the Fundamental

Theorem, Part 2, helps us out.

But in this last case, how do we know that the value we calculate is actually a difference of an antiderivative of  $f(t)$ ? That is, if there is no algebraic formula for  $F(t)$ , how do we know that the value we calculate by taking the limit of a Riemann sum with  $f(t)$  is actually  $F(b) - F(a)$ , where  $F(t)$  is an antiderivative of  $f(t)$ ?

The Fundamental Theorem of Calculus, Part 1, provides the answer. It assures us that if  $f(t)$  is

continuous on  $[a, b]$ , then  $f(t)$  has an antiderivative  $F(t)$  that is defined for all  $t$  in  $[a, b]$  and we can use the definite integral, that is, the limit of a Riemann sum, to form  $F(t)$ . Specifically,

$$F(t) = \int_a^t f(u) du . \text{ As a result, we can say for sure that } \int_a^b f(t) dt = F(b) - F(a) , \text{ where } F(t) \text{ is}$$

an antiderivative of  $f(t)$ .

## You Try It: Part IV

Now you try it with the functions in the list that follows. First, calculate the limit of a Riemann sum, and then find an antiderivative and calculate the difference over the interval of summation. Change the limits on the interval (i.e.,  $a$  and  $b$ ) as well as the function definition for each exercise. We show (d) as an example. (Note that because *Maple* has difficulty evaluating some limits symbolically, we use the **evalf()** and **limit()** commands, which gives a numeric estimate of the limit that is precise to the number of digits available in your computer's numeric processor.

a)  $f(t) = 3t^2 - 6$  for  $(-1 \leq t) \leq 3$

b)  $f(t) = 2 \sin(t)$  for  $(-\pi \leq t) \leq 2\pi$

c)  $f(t) = \frac{2}{t^2}$  for  $(.5 \leq t) \leq 3$

d)  $f(t) = \sqrt{t}$  for  $(1 \leq t) \leq 4$

Change the functions and ranges below to explore the various behaviors.

```
> f := t -> sqrt(t):
a := 1:
b := 4:
area_approx := leftsum( f(t), t=a..b, N ):
area:=evalf(limit( area_approx, N = infinity ));
```

area := 4.666666667

Now we use the Fundamental Theorem.

```
> F :=unapply(int(f(t), t),t):
deltaQ := evalf( F(b) - F(a) );
```

deltaQ := 4.666666667

## Part V: Calculus Gives Birth To New Functions

### The Natural Log Function

As we indicated before, many functions have antiderivatives that cannot be expressed with

algebraic formulas. One example is the function  $f(x) = \frac{1}{x}$ . It is important to realize that while

the antiderivatives of this function cannot be represented by an algebraic formula, an antiderivative does, nonetheless, exist. We can use the Fundamental Theorem, Part 1, to define its antiderivatives, thus giving birth to a new function that can only be defined using calculus.

Specifically, an antiderivative of  $f(x) = \frac{1}{x}$  is defined as  $F(x) = \int_1^x \frac{1}{u} du$  for  $x > 0$ . Since this

$F(x)$  function is so important it is given a special name. It is called the natural log of  $x$  and is denoted by  $\ln x$ . There is a very good reason for calling it a logarithm, that is, it exhibits all of the characteristics of a log function. All of the properties of logarithms hold for  $\ln x$ .

Now the question is how do we evaluate this function? While the Fundamental Theorem, Part 2,

still holds true for  $f(x) = \frac{1}{x}$  and its antiderivative  $\ln x$ , that is,  $\int_a^b \frac{1}{x} dx = \ln(b) - \ln(a)$  it is

not much help in evaluating  $\int_a^b \frac{1}{x} dx$  because there is no algebraic formula for  $\ln x$ .

Consequently, we are forced to resort to its definition to find values for  $\ln x$ . Specifically,  $\ln x$  is

the limit of a Riemann sum. We know that  $\ln(x) = \int_1^x \frac{1}{u} du$  exists because  $f(x) = \frac{1}{x}$  is

continuous on any interval from 1 to  $x$ , provided  $x > 0$ . But how do we represent the values of  $\ln x$  as decimals? Usually we settle for an approximation that is accurate to some specified

precision. One approach is to approximate  $\ln(x) = \int_1^x \frac{1}{u} du$  with a Riemann sum using enough

rectangles to give the desired precision. Let's do it.

We define our own approximate  $\ln x$  function in the following section and call it **approxln(x)**. The variable **n** is the number of rectangles used in the Riemann sum that approximates  $\ln x$ .

```
> approxln := proc(b, n)
    local a, h, c, i;
    a := 1;
    h := (b - a)/n;
    c := i -> a + i*h;
    evalf( sum( (1/c(i))*h, i=0..n - 1 ) );
end;
```

As an exercise, vary the value of  $n$  in the next group of cells to determine how many rectangles are needed to approximate a accurate to three decimal places for  $\left(\frac{1}{10} \leq x\right) \leq 10$  . We start with

$n = 12$  and calculate the error, that is, the difference between our approximate value and that given by **ln(x)** which is *Maple's* command for calculating  $\ln x$ .

```
> n := 12;
approxln(10, n) - ln(10.);
```

0.381747365


```
> approxln(1/10, n) - ln(0.1);
```

0.293252635

Let's compare our approximating function **approxln( )** with *Maple's* **ln(x)** by graphing the two together. First use  $n = 12$ , and then replace it with the value that you found above that ensures three decimal places of precision for  $\left(\frac{1}{10} \leq x\right) \leq 10$  .

```
> n:=12;
plot( [ln(x), approxln(x,n)], x=0.1..10, labels=["x", "y"], color=[red,blue], thickness=[1,1]);
print('The Maple ln(x) is plotted in red and our Riemann sum approximation is plotted in
blue');
```





*The Maple  $\ln(x)$  is plotted in red and our Riemann sum approximation is plotted in blue*

What do you observe?

## You Try It: The Standard Normal Distribution and the Error Function

When you did the You Try It: Part II, you may have thought that the function

$$f(x) = \frac{e^{\left(-\frac{x^2}{2}\right)}}{\sqrt{2\pi}}$$


seemed a bit unusual, but the bell-shaped graph that resulted probably looked

very familiar. This function is a classic example of one that does not have an algebraic form for its antiderivative, and it is called the standard-normal-distribution function, which is used in probability and statistics. Let's examine it in more detail.

> **f := t -> exp(-t^2/2)/sqrt(2\*Pi);**

$$f := t \rightarrow \frac{e^{\left(-\frac{1}{2}t^2\right)}}{\sqrt{2\pi}}$$

> **plot( f(t), t=-5..5, labels=['x', 'y'], thickness=2, color=green, labels=[x,y] );**



Now we consider the definite integral,  $F(x) = \int_{-x}^x f(t) dt = \frac{\int_{-x}^x e^{\left(-\frac{t^2}{2}\right)} dt}{\sqrt{2\pi}}$ , which gives the

area under the bell-shaped curve between  $-x$  and  $x$ . Because  $f(x)$  is an even function, we can

rewrite the integral as  $F(x) = \sqrt{\frac{2}{\pi}} \int_0^x e^{\left(-\frac{t^2}{2}\right)} dt$ .

A function that is closely related to  $F(x)$  is the error function, which is denoted by  $\text{erf } x$  and is

defined as follows:  $\operatorname{erf}(x) = \frac{2}{\sqrt{\pi}} \int_0^x e^{-t^2} dt$ . We leave it as an exercise for you to show that

$F(x) = \operatorname{erf}\left(\frac{x}{\sqrt{2}}\right)$ . We will use  $F(x)$  in the remainder of this module. Since there is no algebraic


formula for the antiderivative of  $f(x)$ , we will approximate values of  $F(x)$  using a Riemann sum with a sufficient number of rectangles to ensure the desired precision.

Using the method that we used for  $\ln x$  in Part V, write *Maple* commands to approximate values of  $F(x)$  to three decimal places of accuracy on the interval  $(-4 \leq x) \leq 4$ . First use a *left-*

*hand* Riemann sum, and then use a midpoint Riemann sum. To test your commands, plot the graph of  $\operatorname{erf}\left(\frac{x}{\sqrt{2}}\right)$  and the graph of your own  $F(x)$  to see if the functions match. (In *Maple*, the

error function is **erf(x)**.) Comment on what you find. We illustrate the Riemann sum with the **middlebox()** command.

> **middlebox( f(t), t=0..-4, 15 );**



Note that the signed areas are negative because we are summing from right to left, and therefore  $h < 0$  and so is  $f(c_i)h$ .

If you need help with this exercise, the following section, contains our solution. Try doing it on your own before you check out what we did.

## Solution

```
> approxf := proc(b, n)
  local a, h, c, i;
  a := 0;
  h := (b - a)/n;
  c := i -> a + (i + 1/2)*h;
  evalf( sqrt(2/Pi) * sum( exp(-c(i)^2/2)*h, i=0..n - 1 ) );
end;
```


```
> approxf( -0.1, 4 );
```

-0.07965774209

```
> evalf( erf(-0.1/sqrt(2)) );
```

-0.07965567453

```
> plot( [approxf(x,n), erf(x/sqrt(2))-.03], x=-2..2, labels=["x", "y"], color=[red,blue], th
[2,2]);
```



```
>
```

NOTE: The "-.03" is just to allow the overlapping functions to show.

There is one final note. Though the use of a Riemann sum to estimate a definite integral is a very direct approach, it is not very efficient. More efficient methods of estimating definite integrals have been developed, and some of these are studied in the module on numerical integration that is entitled "Riemann, Trapezoids, and Simpson."